Supplementary Material: SAS Proc MIXED Syntax for Evaluating Treatment and Covariate Effects with Partially Nested Data

This document provides syntax to implement models presented in Bauer, Sterba, and Hallfors (under review) for evaluating group-based interventions when control participants are ungrouped (and assumed to be independent). Specifically, SAS Proc MIXED syntax, along with annotated excerpts of accompanying SAS output, is provided for each of the three models fit by Bauer et al. in their empirical demonstration evaluating possible iatrogenic effects of the Reconnecting Youth (RY) preventive intervention program.

Data Description.

In the RY effectiveness trial, students were individually assigned to one of three study arms. In the Treatment Arm (RY), high-risk participants received RY treatment *administered in groups* (i.e. RY classes) composed by the experimenter. In the Control Arm (Control), high-risk participants were left *ungrouped*. In the Typical Arm (Typical), low-risk participants were also left *ungrouped*, serving as additional controls. The dependent variable in these analyses was deviant peer bonding (DPB) measured post-treatment. Participants were obtained from 9 high schools, and school was treated as a fixed factor in each of the fitted models (discussion on this point is provided in the original manuscript). As shown in the 6th column of Table A in the next section, Control, Typical, and Treatment subjects could come from the same or different schools. Other covariates common to each study arm were pre-treatment DPB, gender, age, and ethnicity. Regarding ethnicity, the sample was 6.5% Caucasian (N=107), 48.0% Hispanic (N=788), 11.9% African American (N=195), 26.5% Asian American (N=434), and 7.1% American Indian (N=117). Covariates unique to the RY condition were absences from the RY class, percent of students within the class that were female, and the average age of the class members.

Data Preparation.

Two data preparation steps are required specifically for these analyses. First, a group ID variable is required for all participants. A unique value must be assigned to each *group* in the treatment condition (i.e., each RY class) and to each *participant* in the ungrouped conditions. Table A shows an (artificial) section of the dataset with these group IDs assigned in column 1. Here is example SAS code for assigning group IDs, where we assume the user has a pre-existing identification variable called *treatmentgroup*, which uniquely labels each group in which treatment was delivered, and another called *studentID*, which uniquely identifies the individual participants (without using the same numbers or labels as the *treatmentgroup* variable), and a preexisting character variable called *studyarm*, which specifies whether a given participant is from the Treatment, Control, or Typical arm of the study.

```
data RYdat; set RYdat;
 GroupID=.;
 if studyarm="Treatment" then GroupID=treatmentgroup;
 if studyarm="Control" then GroupID=studentID;
 if studyarm="Typical" then GroupID=studentID;
run;
```

Second, the values of any covariates relevant only for the treated (grouped) study arm (e.g., absences, percent female, and average age of class members) need to be set to an arbitrary non-missing value (e.g. -999) for individuals in the non-grouped study arm(s). This is shown in column 5 of Table A. Here is example SAS code for assigning non-grouped individuals arbitrary non-missing values for an original group-level covariate called *percentfem* (indicating the number of females in the RY class the student attended).

```
data RYdat; set RYdat;
 perfem=percentfem;
 if studyarm="Control" then perfem=-999;
 if studyarm="Typical" then perfem=-999;
run;
```

Table A.

GroupID	StudentID	Study Arm	Individual Covariate (e.g. *DPBpre*)	Group Covariate (e.g. *perfem*)	School
G1	1	Treatment	3.00	10.86	2
G1	2	Treatment	2.50	10.86	2
G1	3	Treatment	2.87	10.86	2
G1	4	Treatment	1.50	10.86	2
5	5	Control	0.50	-999	9
6	6	Control	0.00	-999	5
7	7	Typical	0.75	-999	9
8	8	Control	0.63	-999	2
9	9	Typical	1.99	-999	1
G2	10	Treatment	1.13	5.50	5
G2	11	Treatment	1.38	5.50	5
G2	12	Treatment	2.50	5.50	5
G2	13	Treatment	0.63	5.50	5
G2	14	Treatment	1.13	5.50	5
G3	15	Treatment	0.50	7.61	2
G3	16	Treatment	1.27	7.61	2
G3	17	Treatment	2.00	7.61	2

Model Fitting.

Model 1a: Evaluation of Treatment Effect with Assumption of Homoscedasticity for Individual Residuals

A key goal of Model 1a in Bauer et al.'s empirical demonstration was to test whether the average level of deviant peer bonding (measured post-treatment; DPBpost) differed between RY, Control and Typical participants, after including school as a fixed-factor control variable. *RY* is a dichotomous predictor indicating whether a student is assigned to RY-treatment (1), versus Control or Typical (0). *Typical* is a dichotomous predictor indicating whether a student is in assigned to Typical (1), versus Control or RY-Treatment (0). *School* is a nominal predictor with 9 categories, one for each school in the design. School is recoded into 8 dummy variables where

School$(c)_{ij}$ indicates whether a student *i* is in school *c* (coded 1) or not (coded 0). The level 1 equation for Model 1a (Equation 51 in Bauer et al.) is:

$$DPB_{ij} = \beta_{0j} + \beta_{1j} RY_{ij} + \beta_{2j} Typical_{ij} + \sum_{c=1}^{8} \beta_{(2+c)j} School(c)_{ij} + r_{ij}$$

In Model 1a, homoscedasticity for the residuals across arms of the study was assumed (i.e., $r_{ij} \sim N(0, \sigma^2)$ for all three conditions), however, this assumption will be relaxed and tested in Model 1b.

The level 2 equations for Model 1a (Equation 52 in Bauer et al.) are:

$$\begin{aligned} \beta_{0j} &= \gamma_{00} \\ \beta_{1j} &= \gamma_{10} + u_{1j} \\ \beta_{2j} &= \gamma_{20} \\ \beta_{(2+c)j} &= \gamma_{(2+c)0} \end{aligned}$$

Substituting the level 2 equations into the level 1 equation, the combined model equation for Model 1a (Equation 53 in Bauer et al.) is:

$$DPB_{ij} = \gamma_{00} + \gamma_{10} RY_{ij} + \gamma_{20} Typical_{ij} + \sum_{c=1}^{8} \gamma_{c0} School(c)_{ij} + u_{1j} RY_{ij} + r_{ij}$$

The syntax needed to fit this model using the MIXED procedure in SAS is shown below, followed by a brief description of the primary statements.

```
proc mixed data=RYDat method=reml covtest;
 class groupID school;
 model DPBpost = RY typical school / solution ddfm=kr;
 random RY / subject = groupID v vcorr;
 contrast 'Cond' RY 1, typical 1/ e;
 contrast 'Site' school .25 -.2 -.2 -.2 .25 -.2 -.2 .25 .25 / e;
run;
```

The **`proc mixed`** statement calls the MIXED procedure. The **`method=REML`** option calls the restricted maximum likelihood estimator for the model. REML is selected because it typically provides less biased estimates of the variance components of the model than full information maximum likelihood, particularly when there are a small number of groups and/or

large number of covariates. The **covtest** option on the **proc mixed** statement requests asymptotic standard errors and Wald-Z tests for covariance parameters (producing a table of output called "Covariance Parameter Estimates"). Note that SAS does 1-tailed tests of variance parameters.

The **class** statement specifies which variables are classification (i.e. nominal) variables. By including school on the **class** statement, SAS will automatically create 8 dummy variables to capture the 9 levels of school in the data, with the last level of school (School 9) set to the reference category. Our Level 2 ID variable (groupID) also needs to be placed on the **class** statement. Optionally, this can be excluded from the class statement *if* the Level 2 ID variable is numeric and the data is pre-sorted by this ID variable prior to calling the MIXED procedure.

The **model** statement is used to indicate the dependent variable and to specify the fixed effects of the model. Our model includes fixed effects of *RY*, *Typical*, and *School* on *DPBpost*, so these variables are included here. By specifying the **solution** option on the **model** statement we request *t*-tests and standard errors for each fixed effect (output into a table called "Solution for Fixed Effects"). By specifying **ddfm=kr,** we request the standard errors be computed using the Kacker and Harville (1984) approximation, and degrees of freedom be computed according to Kenward and Rogers (1997) method, as described in Bauer, Sterba and Hallfors (under review).

The **random** statement is used to specify the random effects of the model. Thus, on this statement, we list predictors with random effects, i.e., effects that vary randomly across level-2 sampling units. For our model, the only variable with a random effect is *RY*, so this is indicated here. By putting *RY* on the random statement we allow the effect of RY treatment on DPBpost to vary over RY classes and we account for within-class dependence in DPBpost scores. The

level-2 sampling units are identified to SAS using the **subject** option. Here, **subject=** *GroupID*. By specifying the **v** and **vcorr** options on the random statement, we request that the model-implied within-class covariance and correlation matrices for DPBpost be output, conditional on the fixed effects of the model (this is primarily done as a shortcut for calculating the intraclass correlation).

The **contrast** statement tests composite hypotheses involving linear combinations of fixed and/or random effects and outputs the *F*-test of each linear contrast to a table in the output titled "Contrasts." Here, the first contrast statement is used to test whether there is a significant DBP mean difference between students assigned to the three study arms. The second contrast statement is used to test whether there is a significant DPB mean difference between the 4 schools at site A (weighted .25) and the 5 schools at site B (weighted -.2). By specifying the **e** option on the **contrast** statement, we request that the matrix of coefficients for the contrast be output so we can verify that the contrasts were specified as intended. These matrices are found in the "Coefficients" table of the output window.

Here, we provide a subset of the output produced by SAS for Model 1a. Portions of output that can be matched to values in the first column of Table 1 and to interpretations on page 29-30 of Bauer, Sterba, and Hallfors (under review) are indicated in bold font. First we show the "Solution for Fixed Effects" table from the output window:

Solution for Fixed Effects

Effect	school	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept		1.1113	0.06833	1167	16.26	<.0001
RY		**0.1880**	0.07156	**68.3**	**2.63**	0.0106
Typical		**-0.3690**	0.05281	**1432**	**-6.99**	<.0001
school	1	0.4794	0.09903	1103	4.84	<.0001
school	2	0.7387	0.09311	752	7.93	<.0001
school	3	0.7150	0.1256	767	5.69	<.0001
school	4	0.7485	0.09313	933	8.04	<.0001
school	5	0.2214	0.09754	921	2.27	0.0234
school	6	0.7190	0.09051	868	7.94	<.0001
school	7	0.8249	0.1240	1284	6.65	<.0001
school	8	0.4376	0.09087	744	4.82	<.0001
school	9	0	.	.	.	.

Each fixed effect is labeled by the variable to which it refers, and is accompanied by a *t*-test, degrees of freedom, and *p*-value. The fixed effect of *RY* represents the average difference in post-treatment DPB scores for individuals in the RY condition relative to the Control condition, whereas the fixed effect of *Typical* indicates the average difference in DPB scores for individuals in the Typical condition relative to the Control condition. Each school effect represents the mean difference in DPB scores between students of a particular school (1 through 8) relative to school 9, the reference school. A joint test of these eight fixed effects, representing the main effect of School, is provided by SAS in the "Type 3 Tests of Fixed Effects" table of output:

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
RY	1	68.3	6.90	0.0106
Typical	1	1432	48.82	<.0001
school	**8**	**891**	**15.95**	<.0001

Note that the single degree of freedom tests reported here are redundant with the *t*-tests in the previous table of output.

Additionally, the "Contrasts" table in the output window provides *F*-test of our two linear contrasts, each identified with the label we placed in single quotes in the **`contrast`** statement.

Contrasts

Label	Num DF	Den DF	F Value	Pr > F
Cond	**2**	**132**	**39.48**	<.0001
Site	**1**	**936**	**80.75**	<.0001

The *F*-test of the 'Cond' contrast indicates that there is an overall mean difference in DPB between individuals in the RY, Control and Typical conditions. This represents a joint test of the *RY* and *Typical* fixed effects in the "Solution for Fixed Effects" table presented previously. The *F*-test of the 'Site' contrast indicates that there is a significant overall mean difference in DPB for individuals from schools located at Site A versus individuals from schools located at Site B.

The estimated variance components of the model are shown in the "Covariance Parameter Estimates" table below:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
RY	groupID	**0.05263**	0.03474	**1.51**	0.0649
Residual		**0.7888**	0.02951	**26.73**	<.0001

In general, each random effect variance is labeled by the variable to which it refers. In this case, the variance labeled 'RY' represents the variance of the random slope for the regression of DPB on RY treatment, which is nearly significant by the 1-tailed *z*-test, suggesting some variability across treatment groups within the RY arm (controlling for school effects). The variance of the individual (Level 1) residuals is labeled 'Residual'. This model assumes that the variance of these residuals is constant across the three arms of the study (implying that there is no homogenization of behavior within RY groups), an assumption that will later be tested. Using

the formula in Equation (22) of Bauer, Sterba & Hallfors (under review), the variance components obtained from this model result in an ICC for the RY condition of .06. This value could also be obtained directly from the off-diagonal elements of the SAS output produced by the **`vcorr`** option, shown below.

Estimated V Correlation Matrix for groupID 10001

Row	Col1	Col2	Col3	Col4	Col5	Col6	Col7	Col8
1	1.0000	0.06255	0.06255	0.06255	0.06255	0.06255	0.06255	0.06255
2	0.06255	1.0000	0.06255	0.06255	0.06255	0.06255	0.06255	0.06255
3	0.06255	0.06255	1.0000	0.06255	0.06255	0.06255	0.06255	0.06255
4	0.06255	0.06255	0.06255	1.0000	0.06255	0.06255	0.06255	0.06255
5	0.06255	0.06255	0.06255	0.06255	1.0000	0.06255	0.06255	0.06255
6	0.06255	0.06255	0.06255	0.06255	0.06255	1.0000	0.06255	0.06255
7	0.06255	0.06255	0.06255	0.06255	0.06255	0.06255	1.0000	0.06255
8	0.06255	0.06255	0.06255	0.06255	0.06255	0.06255	0.06255	1.0000

A final piece of output that will become relevant shortly is the -2*log-likelihood (or model deviance), which is produced by SAS in the "Fit Statistics" table:

Fit Statistics

-2 Res Log Likelihood	**3888.0**
AIC (smaller is better)	3892.0
AICC (smaller is better)	3892.0
BIC (smaller is better)	3902.4

We require the -2*log-likelihood, here reported as 3888.0, in order to evaluate the assumption of homoscedasticity of the individual residuals with our next model.

Model 1b: Allowing for Heteroscedasticity of the Individual Residuals

In Model 1b we allow the variance of the individual residuals to differ across the grouped and ungrouped conditions (i.e. RY condition versus Control and Typical conditions) to evaluate whether grouping participants results in the homogenization of behavior. To accomplish this, we will use the **`repeated`** statement with the **`group`** option in PROC MIXED.

```
proc sort data=RYDat; by RY; run;

proc mixed data=RYDat method=reml covtest;
 class groupID school;
 model DPBpost = RY typical school / solution ddfm=kr;
 random RY / subject = groupID v vcorr;
 repeated / group=RY;
run;
```

In general, the **repeated** statement provides access to the variances and covariances of the individual (level 1) residuals. In Model 1a, these residuals were assumed to be independent and of constant variance across the grouped (Treatment) and ungrouped (Control and Typical) arms of the study. With Model 1b, we relax the assumption of constant variance, requesting that a different variance be estimated for the residuals of individuals in the RY condition relative to the two ungrouped conditions. This is accomplished by including the **group=RY** option on the **repeated** statement.

The primary output of interest for this model is the table of "Covariance Parameter Estimates", which provides our new residual variance estimates:

Covariance Parameter Estimates

Cov Parm	Subject	Group	Estimate	Standard Error	Z Value	Pr Z
RY	groupID		0.04980	0.03536	1.41	0.0795
Residual		Group 1	**0.7824**	0.03286	23.81	<.0001
Residual		Group 2	**0.8133**	0.06698	12.14	<.0001

As can be seen, the residual variance for the *RY* condition (labeled 'Group 2') is higher than the residual variance for the other two conditions (labeled 'Group 1'), in contradiction to the hypothesis of within-group homogenization. (Note that SAS labels each level of the variable specified in the group option in the order that it comes to it in the data processing). We are also interested in the -2*log-likelihood of the model which, when contrasted with the -2*log-

likelihood of Model 1a, will provide a likelihood ratio test of the assumption of homoscedasticity for the individual residuals. We obtain this from the "Fit Statistics" table, shown here:

Fit Statistics	
-2 Res Log Likelihood	**3887.8**
AIC (smaller is better)	3893.8
AICC (smaller is better)	3893.8
BIC (smaller is better)	3909.4

The difference in fit between the two models, as measured by the -2*log-likelihood, follows a chi-square distribution with degrees of freedom equal to the difference in the number of parameters (in this case, one). The likelihood ratio test can be computed directly in SAS using the following syntax:

```
data LRT;
 dev1 =  3888.0;
 dev2 =  3887.8;
 chi = dev1-dev2;
 p = 1-probchi(chi,1);
run;
proc print data=LRT; run;
```

The **`probchi`** function returns the probability of obtaining a likelihood ratio less than or equal to the one we obtained. Subtracting from 1 then yields the probability of finding a likelihood ratio this large or larger, under the null hypothesis of no difference. The result is

Obs	dev1	dev2	chi	p
1	3888	3887.8	0.2	0.65472

We thus find that the individual residual variances are not sufficiently different between the study arms to warrant rejection of the homoscedastic model (Model 1a).

Model 2: Adding Common Covariates

Model 2 includes four additional level-1 fixed effect covariates measured in all three study arms: baseline delinquency (*DPBpre*), *ethnicity*, *sex*, and age (*ageyrs*). We will call these

"common" covariates to emphasize that they are measured in all study arms. Model 2 assumes that the relationship between each common covariate and DPBpost is constant over treatment groups. Since these covariates enter the model only through the addition of fixed effects, we forgo writing out the model equations here, as this becomes somewhat tedious. The additional fixed effects of the covariates are produced simply by adding the covariates to the **`model`** statement, as shown below.

```
proc mixed data=RYdat method=reml covtest;
 class groupID school ethnicity sex;
 model DPBpost = RY typical School DPBpre ethnicity ageyrs sex / solution
                 ddfm=kr;
 random RY / subject = groupID v vcorr;
 contrast 'Cond' RY 1, typical 1/ e;
 contrast 'Site' school .25 -.2 -.2 -.2 .25 -.2 -.2 .25 .25 / e;
run;
```

Note that the nominal covariates, ethnicity and sex, are also added to the **`class`** statement to indicate that they are classification variables. The rest of the syntax is identical to Model 1a.

Selections from the output are shown below. Bold portions of output from Model 2 can be matched to values in the second column of Table 1 and to interpretations on page 30 of the text. We first consider the solution for the fixed effects:

Solution for Fixed Effects

Effect	ethnicity	school	gender	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept				1.2295	0.3889	1451	3.16	0.0016
RY				**0.1382**	0.05691	**79.4**	**2.43**	0.0175
Typical				**-0.1063**	0.04847	**1428**	**-2.19**	0.0285
school		1		0.2367	0.09029	1113	2.62	0.0089
school		2		0.3077	0.1009	931	3.05	0.0023
school		3		0.2424	0.1268	831	1.91	0.0562
school		4		0.3855	0.1025	1096	3.76	0.0002
school		5		0.04995	0.08612	864	0.58	0.5621
school		6		0.3109	0.09730	1108	3.20	0.0014
school		7		0.4771	0.1285	1372	3.71	0.0002
school		8		0.1702	0.07922	582	2.15	0.0321
school		9		0	.	.	.	.
DPBpre				**0.4958**	0.02326	**1444**	**21.31**	<.0001
ethnicity	Ameri			0.02161	0.1132	1455	0.19	0.8487
ethnicity	Asian			-0.2346	0.08997	1453	-2.61	0.0092
ethnicity	Black			-0.01719	0.1112	1455	-0.15	0.8771
ethnicity	Latin			-0.1569	0.09773	1453	-1.61	0.1085
ethnicity	White			0	.	.	.	.
ageyrs				**-0.03498**	0.02499	**1450**	**-1.40**	0.1618
sex			1.00	**0.06452**	0.04098	**1455**	**1.57**	0.1157
sex			2.00	0	.	.	.	.

For the covariates indicated in the **`class`** statement, multi-degree of freedom tests of main effects are also available in the "Type 3 Tests of Fixed Effects" table.

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
RY	1	79.4	5.89	0.0175
Typical	1	1428	4.81	0.0285
school	**8**	**781**	**3.03**	0.0023
DPBpre	1	1444	454.27	<.0001
ethnicity	**4**	**1452**	**4.07**	0.0028
ageyrs	1	1450	1.96	0.1618
sex	1	1455	2.48	0.1157

Overall, only the *t*-test of the fixed effect of *DBPpre* in the "Solutions for Fixed Effects" table and the overall *F*-test of the fixed effect of *ethnicity* and *school* in the "Type 3 Tests of Fixed Effects" table uniquely explain additional variability in DBPpost. The differences between the three study arms, however, are maintained after controlling for the covariates. Our two linear contrasts also remain significant as shown below.

Contrasts

Label	Num DF	Den DF	F Value	Pr > F
Cond	2	149	8.48	**0.0003**
Site	1	1133	12.49	**0.0004**

The other output of interest from this model are the variance component estimates, shown here:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
RY	groupID	**0.01420**	0.01995	0.71	0.2383
Residual		**0.5925**	0.02225	26.63	<.0001

As can be seen by comparison to the variance components obtained from Model 1a, the inclusion of the four pre-existing difference covariates explains almost all of the across-group variability in the effect of RY treatment on DPB. This is evidenced by the nonsignificant, near-zero estimate for the variance of the random slope of RY-treatment on DPB (labeled *RY* in the "Covariance Parameter Estimates" output table.) This represents a decrease in the variance of the random slope from .05263 in Model 1a to .01420 here in Model 2.

Model 3: Adding Covariates Unique to Grouped Condition

The syntax for Model 3 adds two group-level predictors to explain why treatment may have been more detrimental for some RY-treatment groups than others. They are the mean age of the group (*meanage*), and the percentage of females in the group (*perfem*). In addition, the individual level predictor, number of absences from the RY class, was included to assess possible dosage effects. These predictors are only relevant for members of RY classes and are, in essence, undefined for participants in the other study arms. For this reason, they should only

impact the DPB scores of students in the RY study arm and should have no effect for the other two study arms. To meet these constraints, these predictors are entered through interactions with the RY variable (see extended discussion in Bauer, Sterba & Hallfors, under review). No main effects are included because the effect is necessarily null in the control conditions. Additionally, it is worth repeating that these covariates are coded -999 for all participants in the control or typical conditions (an arbitrary value that is *not* declared as a missing data code) so that these participants will not be listwise deleted from the analysis.

The syntax for fitting the model is:

```
proc mixed data=RYdat method=reml covtest;
 class groupID school ethnicity sex;
 model DPBpost = RY typical School DPBpre ethnicity ageyrs sex
       RY*meanage RY*absences RY*perfem/ solution ddfm=kr;
 random RY / subject = groupID v vcorr;
 contrast 'Cond' RY 1, typical 1;
 contrast 'Site' school .25 -.2 -.2 -.2 .25 -.2 -.2 .25 .25 / e;
run;
```

The syntax for Model 3 is identical to the syntax from Model 2 except for the three new fixed effects, *RY*meanage*, *RY*perfem*, and *RY*absences*, that have each been added to the **`model`** statement. These new fixed effects are cross-products that can be indicated directly within the **`model`** statement; they need not be computed ahead of time in a data step. Notice that we did not enter the main effects of these predictors on the **`model`** statement.

The estimates for the fixed effects are shown here, with bolded entries appearing in column 3 of table 1:

Solution for Fixed Effects

Effect	ethnicity	school	gender	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept				1.1307	0.3942	1423	2.87	0.0042
RY				0.1392	0.05757	64.7	2.42	0.0184
Typical				-0.1066	0.04844	1405	-2.20	0.0279
school		1		0.2400	0.09039	1049	2.65	0.0081
school		2		0.3022	0.1027	1074	2.94	0.0033
school		3		0.2464	0.1271	764	1.94	0.0530
school		4		0.3841	0.1029	1030	3.73	0.0002
school		5		0.04509	0.08627	801	0.52	0.6014
school		6		0.3148	0.09740	1025	3.23	0.0013
school		7		0.4797	0.1285	1322	3.73	0.0002
school		8		0.1641	0.07969	532	2.06	0.0399
school		9		0	.	.	.	.
DPBpre				0.4922	0.02347	1413	20.97	<.0001
ethnicity	Ameri			0.001993	0.1139	1429	0.02	0.9860
ethnicity	Asian			-0.2448	0.09021	1429	-2.71	0.0067
ethnicity	Black			-0.03326	0.1119	1429	-0.30	0.7663
ethnicity	Latin			-0.1623	0.09842	1429	-1.65	0.0994
ethnicity	White			0	.	.	.	.
ageyrs				-0.02721	0.02538	1419	-1.07	0.2839
sex			1.00	0.05628	0.04158	1409	1.35	0.1761
sex			2.00	0	.	.	.	.
RY*meanage				**-0.1362**	0.1336	33.4	**-1.02**	**0.3150**
RY*absences				**0.001584**	0.003083	662	**0.51**	**0.6076**
RY*perfem				**-0.00128**	0.002655	34.7	**-0.48**	**0.6327**

Considering first the group-level predictors, we can see that having more girls (*RY*perfem)* and older students in the class (*RY*meanage*) slightly decreases DPB post scores, though neither effect is significant. Additionally, more absences (*RY*absences)* are associated with higher DPBpost scores. Other output (i.e., covariance parameters, contrasts, and multi-degree of freedom tests) is identical in form to Models 1a and 2 and so is not presented here.